# Maintainer: Jan de Groot <jgc@archlinux.org>
# Contributor: Chaiwat Suttipongsakul <cwt114@gmail.com>

pkgname=libdatrie
pkgver=0.2.13
pkgrel=2
pkgdesc="Implementation of double-array structure for representing trie, as proposed by Junichi Aoe."
url="https://linux.thai.net/projects/datrie"
license=('LGPL')
arch=('x86_64')
depends=('glibc')
provides=('libdatrie.so')
options=('!emptydirs' 'debug')
source=(https://linux.thai.net/pub/thailinux/software/libthai/${pkgname}-${pkgver}.tar.xz)
md5sums=('e26b5aa008b5f3588ab38d2dce9e9325')
sha256sums=('12231bb2be2581a7f0fb9904092d24b0ed2a271a16835071ed97bed65267f4be')

build() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  ./configure --prefix=/usr --disable-static
  make
}

package() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  make DESTDIR="${pkgdir}" install -j1
}
